# Disjunctive Delimited Control

Alexander Vandenbroucke $^1$  and Tom Schrijvers $^2$ 

Standard Chartered alexander.vandenbroucke@sc.com KU Leuven tom.schrijvers@kuleuven.be

Abstract. Delimited control is a powerful mechanism for programming language extension which has been recently proposed for Prolog (and implemented in SWI-Prolog). By manipulating the control flow of a program from inside the language, it enables the implementation of powerful features, such as tabling, without modifying the internals of the Prolog engine. However, its current formulation is inadequate: it does not capture Prolog's unique non-deterministic nature which allows multiple ways to satisfy a goal.

This paper fully embraces Prolog's non-determinism with a novel interface for disjunctive delimited control, which gives the programmer not only control over the sequential (conjunctive) control flow, but also over the non-deterministic control flow. We provide a meta-interpreter that conservatively extends Prolog with delimited control and show that it enables a range of typical Prolog features and extensions, now at the library level: findall, cut, branch-and-bound optimisation, probabilistic programming, ...

**Keywords:** delimited control, disjunctions, Prolog, meta-interpreter, branch-and-bound

#### 1 Introduction

Delimited control is a powerful programming language mechanism for control flow manipulation that was developed in the late '80s in the context of functional programming [5,2]. Schrijvers et al. [10] have recently ported this mechanism to Prolog.

This port enables powerful applications in Prolog, such as high-level implementations of both tabling [3] and algebraic effects & handlers [7]. Yet, at the same time, their work leaves much untapped potential, as it fails to recognise the unique nature of Prolog when compared to functional and imperative languages that have previously adopted delimited control.

Indeed, computations in other languages have only one *continuation*, i.e., one way to proceed from the current point to a result. In contrast, at any point in a Prolog continuation, there may be multiple ways to proceed and obtain a result. More specifically, we can distinguish 1) the success or *conjunctive* continuation which proceeds with the current state of the continuation; and 2) the failure or

disjunctive continuation which bundles the alternative ways to proceed, e.g., if the conjunctive continuation fails.

The original delimited control only accounts for one continuation, which Schrijvers et al. have unified with Prolog's conjunctive continuation. More specifically, for a given subcomputation, they allow to wrest the current conjunctive continuation from its track, and to resume it at leisure, however many times as desired. Yet, this entirely ignores the disjunctive continuation, which remains as and where it is.

In this work, we adapt delimited control to embrace the whole of Prolog and capture both the conjunctive and the disjunctive continuations. This makes it possible to manipulate Prolog's built-in search for custom search strategies and enables clean implementations of, e.g., findall/3 and branch-and-bound. This new version of delimited control has an executable specification in the form of a meta-interpreter (Section 3), that can run both the above examples, amongst others.

#### 2 Overview and Motivation

#### 2.1 Background: Conjunctive Delimited Control

In earlier work, Schrijvers et al. [10] have introduced a Prolog-compatible interface for delimited control that consists of two predicates: reset/3 and shift/1. To paraphrase their original description, reset(Goal, ShiftTerm, Cont) executes Goal, and,

A: Maybe this shouldn't be inlined (We have enough space.)

- 1. if Goal fails, reset/3 also fails;
- if Goal succeeds, then reset/3 also succeeds and unifies Cont and ShiftTerm with 0;
- 3. if Goal calls shift(Term), then the execution of Goal is suspended and reset/3 succeeds immediately, unifying ShiftTerm with Term and Cont with the remainder of Goal.

The shift/reset pair resembles the more familiar catch/throw predicates, with the following differences: shift/1 does not copy its argument (i.e., it does not refresh the variables), it does not delete choice points, and also communicates the remainder of Goal to reset/3.

Obliviousness to Disjunctions This form of delimited control only captures the conjunctive continuation. For instance reset((shift(a),G1),Term,Cont) captures in Cont goal G1 that appears in conjunction to shift(a). In a low-level operational sense this corresponds to delimited control in other (imperative and functional) languages where the only possible continuation to capture is the computation that comes sequentially after the shift. Thus this approach is very useful for enabling conventional applications of delimited control in Prolog.

In functional and imperative languages delimited control can also be characterised at a more conceptual level as capturing the entire remainder of a

computation. Indeed, in those languages the sequential continuation coincides with the entire remainder of a computation. Yet, the existing Prolog approach fails to capture the entire remainder of a goal, as it only captures the conjunctive continuation and ignores any disjunctions. This can be illustrated by the reset((shift(a),G1;G2),Term,Cont) which only captures the conjunctive continuation G1 in Cont and not the disjunctive continuation G2. In other words, only the conjunctive part of the goal's remainder is captured.

This is a pity because disjunctions are a key feature of Prolog and many advanced manipulations of Prolog's control flow involve manipulating those disjunctions in one way or another. Indeed, we will see several examples of such manipulations in the next sections.

#### 2.2 Delimited Continuations with Disjunction

This paper presents an approach to delimited control for Prolog that is in line with the conceptual view that the whole remainder of a goal should be captured, including in particular the disjunctive continuation.

For this purpose we modify the reset/3 interface, where depending on Goal, reset(Pattern, Goal, Result) has three possible outcomes:

1. If Goal fails, then the reset succeeds and unifies Result with failure. For instance,

```
?- reset(_,fail,Result).
Result = failure.
```

2. If Goal succeeds, then Result is unified with success(PatternCopy, DisjCont) and the reset succeeds. Here DisjCont is a goal that represents the disjunctive remainder of Goal. For instance,

```
?- reset(X,(X = a; X = b),Result).
X = a, Result = success(Y,Y = b).
```

Observe that, similar to findall/3, the logical variables in DisjCont have been renamed apart to avoid interference between the branches of the computation. To be able to identify any variables of interest after renaming, we provide PatternCopy as a likewise renamed-apart copy of Pattern.

3. If Goal calls shift(Term), then the reset succeeds and Result is unified with shift(Term,ConjCont,PatternCopy,DisjCont). This contains in addition to the disjunctive continuation also the conjunctive continuation. The latter is not renamed apart and can share variables with Pattern and Term. For instance,

```
?- reset(X,(shift(t),X = a; X = b),Result).
Result = shift(t,X = a, Y, Y = b).
```

Note that  $\mathtt{reset}(P,G,R)$  always succeeds if R is unbound and never leaves choicepoints.

Encoding findall/3 Section 4 presents a few larger applications, but our encoding of findall/3 with disjunctive delimited control already gives some idea of the expressive power:

```
findall(Pattern,Goal,List) :-
   reset(Pattern,Goal,Result),
   findall_result(Result,Pattern,List).

findall_result(failure,_,[]).
  findall_result(success(PatternCopy,DisjCont),Pattern,List) :-
   List = [Pattern|Tail],
   findall(PatternCopy,DisjCont,Tail).
```

This encoding is structured around a reset/3 call of the given Goal followed by a case analysis of the result. Here we assume that shift/1 is not called in Goal, which is a reasonable assumption for plain findall/3.

Encoding !/O Our encoding of the !/O operator illustrates the use of shift/1:

```
cut :- shift(cut).
scope(Goal) :-
   copy_term(Goal,Copy),
   reset(Copy,Copy,Result),
   scope_result(Result,Goal,Copy).

scope_result(failure,_,_) :-
   fail.
scope_result(success(DisjCopy,DisjGoal),Goal,Copy) :-
   Goal = Copy.
scope_result(success(DisjCopy,DisjGoal),Goal,Copy) :-
   DisjCopy = Goal,
   scope(DisjGoal).
scope_result(shift(cut,ConjGoal,DisjCopy,DisjGoal),Goal,Copy) :-
   Copy = Goal,
   scope(ConjGoal).
```

The encoding provides cut/0 as a substitute for !/0. Where the scope of regular cut is determined lexically, we use scope/1 here to define it dynamically. For instance, we encode

```
 \begin{array}{c} p(X,Y) := q(X), !, r(Y). \\ p(4,2). \end{array} \text{ as } \begin{array}{c} p(X,Y) := scope(p_aux(X,Y)). \\ p_aux(X,Y) := q(X), cut, r(Y). \\ p_aux(4,2). \end{array}
```

The logic of cut is captured in the definition of scope/1; all the cut/0 predicate does is request the execution of a cut with shift/1.

In scope/1, the Goal is copied to avoid instantiation by any of the branches. The copied goal is executed inside a reset/3 with the copied goal itself as the pattern. The scope\_result/3 predicate handles the result:

- failure propagates with fail;
- success creates a disjunction to either unify the initial goal with the now instantiated copy to propagate bindings, or to invoke the disjunctive continuation;
- shift(cut) discards the disjunctive continuation and proceeds with the conjunctive continuation only.

### 3 Meta-Interpreter Semantics

We provide an accessible definition of disjunctive delimited control in the form of a meta-interpreter. Broadly speaking, it consists of two parts: the core interpreter, and a top level predicate to initialise the core and interpret the results.

#### 3.1 Core Interpreter

Figure 1 defines the interpreter's core predicate, eval(Conj, PatIn, Disj, PatOut, Result). It captures the behaviour of reset(Pattern, Goal, Result) where the goal is given in the form of a list of goals, Conj, together with the alternative branches, Disj. The latter is renamed apart from Conj to avoid conflicting instantiations.

The pattern that identifies the variables of interest (similar to findall/3) is present in three forms. Firstly, PatIn is an input argument that shares the variables of interest with Conj (but not with Disj). Secondly, PatOut outputs the instantiated pattern when the goal succeeds or suspends on a shift/1. Thirdly, the alternative branches Disj are of the form alt(BranchPatIn,BranchGoal) with their own copy of the pattern.

When the conjunction is empty (1–4), the output pattern is unified with the input pattern, and success/2 is populated with the information from the alternative branches.

When the first conjunct is true/0 (5–6), it is dropped and the meta-interpreter proceeds with the remainder of the conjunction. When it is a composite conjunction (G1,G2) (7–8), the individual components are added separately to the list of conjunctions.

When the first conjunct is fail/0 (9–10), the meta-interpreter backtracks explicitly by means of auxiliary predicate backtrack/3.

```
backtrack(Disj,PatOut,Result) :-
    ( empty_alt(Disj) ->
        Result = failure
    ; Disj = alt(BranchPatIn,BranchGoal) ->
        empty_alt(EmptyDisj),
        eval([BranchGoal],BranchPatIn,EmptyDisj,PatOut,Result)
```

```
eval([],PatIn,Disj,PatOut,Result) :- !,
1
        PatOut = PatIn,
2
        Disj
                    = alt(BranchPatIn,BranchGoal),
3
        Result
                    = success(BranchPatIn,BranchGoal).
     eval([true|Conj],PatIn,Disj,PatOut,Result) :- !,
         eval(Conj,PatIn,Disj,PatOut,Result).
     eval([(G1,G2)|Conj],PatIn,Disj,PatOut,Result) :-!,
         eval([G1,G2|Conj],PatIn,Disj,PatOut,Result).
     eval([fail|_Conj],_,Disj,PatOut,Result) :- !,
9
         backtrack(Disj,PatOut,Result).
10
     eval([(G1;G2)|Conj],PatIn,Disj,PatOut,Result) :- !,
11
         copy_term(alt(PatIn,conj([G2|Conj])),Branch),
         disjoin(Branch, Disj, New Disj),
13
         eval([G1|Conj],PatIn,NewDisj,PatOut,Result).
14
     eval([conj(Cs)|Conj],PatIn,Disj,PatOut,Result) :- !,
15
         append(Cs,Conj,NewConj),
16
         eval(NewConj,PatIn,Disj,PatOut,Result).
17
     eval([shift(Term)|Conj],PatIn,Disj,PatOut,Result) :- !,
         PatOut = PatIn,
19
                    = alt(BranchPatIn,Branch),
20
                    = shift(Term,conj(Conj),BranchPatIn,Branch).
         Result
21
     eval([reset(RPattern,RGoal,RResult)|Conj],PatIn,Disj,PatOut,Result):- !,
22
         copy_term(RPattern-RGoal, RPatIn-RGoalCopy),
23
         empty_alt(RDisj),
         eval([RGoalCopy],RPatIn,RDisj,RPatOut,RResultFresh),
         eval([RPattern=RPatOut, RResult=RResultFresh|Conj]
26
              ,PatIn,Disj,PatOut,Result).
27
     eval([Call|Conj],PatIn,Disj,PatOut,Result) :- !,
28
         findall(Call-Body, clause(Call, Body), Clauses),
29
         ( Clauses = [] -> backtrack(Disj,PatOut,Result)
          disjoin_clauses(Call,Clauses,ClausesDisj),
31
           eval([ClausesDisj|Conj],PatIn,Disj,PatOut,Result)
32
33
```

Fig. 1. Meta-Interpreter Core

```
empty_alt(alt(_,fail)).
```

If there is no alternative branch, it sets the Result to failure. Otherwise, it resumes with the alternative branch. Note that by managing its own backtracking, eval/5 is entirely deterministic with respect to the meta-level Prolog system.

When the first conjunct is a disjunction (G1;G2) (11-14), the meta-interpreter adds (a renamed apart copy of) (G2,Conj) to the alternative branches with disjoin/3 and proceeds with [G1|Conj].

```
disjoin(alt(_,fail),Disjunction,Disjunction) :- !.
disjoin(Disjunction,alt(_,fail),Disjunction) :- !.
disjoin(alt(P1,G1),alt(P2,G2),Disjunction) :-
Disjunction = alt(P3, (P1 = P3, G1; P2 = P3, G2)).
```

Note that we have introduced a custom built-in conj(Conj) that turns a list of goals into an actual conjunction. It is handled (15–17) by prepending the goals to the current list of conjuncts, and never actually builds the explicit conjunction.

When the first goal is shift(Term) (18-21), this is handled similarly to an empty conjunction, except that the result is a shift/4 term which contains Term and the remainder of the conjunction in addition the branch information.

When the first goal is a reset (RPattern, RGoal, RResult) (22-27), the meta-interpreter sets up an isolated call to eval/5 for this goal. When the call returns, the meta-interpreter passes on the results and resumes the current conjunction Conj. Notice that we are careful that this does not result in meta-level failure by meta-interpreting the unification.

Finally, when the first goal is a call to a user-defined predicate (28–33), the meta-interpreter collects the bodies of the predicate's clauses whose head unifies with the call. If there are none, it backtracks explicitly. Otherwise, it builds an explicit disjunction with disjoin\_clauses, which it pushes on the conjunction stack.

```
disjoin_clauses(_G,[],fail) :- !.
disjoin_clauses(G,[GC-Clause],(G=GC,Clause)) :- !.
disjoin_clauses(G,[GC-Clause|Clauses],((G=GC,Clause); Disj)) :-
    disjoin_clauses(G,Clauses,Disj).
```

An example execution trace of the interpreter can be found in C.

Toplevel The toplevel(Goal)-predicate initialises the core interpreter with a conjunction containing only the given goal, the pattern and pattern copy set to (distinct) copies of the goal, and an empty disjunction. It interprets the result by non-deterministically producing all the answers to Goal and signalling an error for any unhandled shift/1.

```
toplevel(Goal) :-
    copy_term(Goal,GoalCopy),
    PatIn = GoalCopy,
    empty_alt(Disj),
    eval([GoalCopy],PatIn,Disj,PatOut,Result),
    ( Result = success(BranchPatIn,Branch) ->
        ( Goal = PatOut ; Goal = BranchPatIn, toplevel(Branch))
    ; Result = shift(_,_,_,_) ->
        write('toplevel: uncaught shift/1.\n'), fail
    ; Result = failure ->
```

```
bound(V) :- shift(V).

bb(Value,Data,Goal,Min) :-
    reset(Data,Goal,Result),
    bb_result(Result,Value,Data,Min).

bb_result(success(BranchCopy,Branch),Value,Data,Min) :-
    ( Data @< Value -> bb(Data,BranchCopy,Branch,Min)
    ; bb(Value,BranchCopy,Branch,Min)
    ).

bb_result(shift(ShiftTerm,Cont,BranchCopy,Branch),Value,Data,Min) :-
    ( ShiftTerm @< Value ->
        bb(Value,Data,(Cont ; (BranchCopy = Data,Branch)),Min)
    ; bb(Value,BranchCopy,Branch,Min)
    ).

bb_result(failure,Value,_Data,Min) :- Value = Min.
```

Fig. 2. Branch-and-Bound Effect Handler.

```
fail
).
```

## 4 Case Studies

To illustrate the usefulness and practicality of our approach, we present two case studies that use the new reset/3 and shift/1.<sup>3</sup>

#### 4.1 Branch-and-Bound: Nearest Neighbour Search

Branch-and-bound is a well-known general optimisation strategy, where the solutions in certain areas or branches of the search space are known to be bounded. Such branches can be pruned, when their bound does not improve upon a previously found solution, eliminating large swaths of the search space in a single stroke.

We provide an implementation of branch-and-bound (see Figure 2) that is generic, i.e., it is not specialised for any application. In particular it is not specific to nearest neighbour search, the problem on which we demonstrate the branch-and-bound approach here.

The framework requires minimal instrumentation: it suffices to begin every prunable branch with bound(V), where V is a lower bound on the values in the branch.<sup>4</sup>

<sup>&</sup>lt;sup>3</sup> More examples are available at https://people.cs.kuleuven.be/~alexander.vandenbroucke/publications/disjunctive-continuations.tgz.

<sup>&</sup>lt;sup>4</sup> The framework searches for a minimal solution.

```
nn((X,Y),BSP,D-(NX,NY)) :-
    ( BSP = xsplit((SX,SY),Left,Right) ->
        DX is X - SX,
        branch((X,Y), (SX,SY), Left, Right, DX, D-(NX,NY))
    ; BSP = ysplit((SX,SY),Up,Down) ->
        DY is Y - SY,
        branch((X,Y), (SX,SY), Up, Down, DY, D-(NX,NY))
    ).
branch((X,Y), (SX,SY), BSP1, BSP2, D, Dist-(NX,NY)) :-
    ( D < 0 \rightarrow % Find out which partition contains (X,Y).
        TargetPart = BSP1, OtherPart = BSP2, BoundaryDistance is -D
        TargetPart = BSP2, OtherPart = BSP1, BoundaryDistance is D
    ),
    ( nn((X,Y), TargetPart, Dist-(NX,NY))
    ; Dist is (X - SX) * (X - SX) + (Y - SY) * (Y - SY),
      (NX,NY) = (SX,SY)
    ; bound(BoundaryDistance-nil),
      nn((X,Y), OtherPart, Dist-(NX,NY))
run nn((X0,Y0),BSP,(NX,NY)):
    toplevel(bb(10-nil,D-(X,Y),nn((X0,Y0),BSP,D-(X,Y)),_-(NX,NY))).
```

Fig. 3. 2D Nearest Neighbour Search with Branch-and-Bound.

- 1. If the Goal succeeds normally (i.e., Result is success), then Data contains a new solution, which is only accepted if it is an improvement over the existing Value. The handler then tries the next Branch.
- 2. If the Goal calls bound(V), V is compared to the current best Value:
  - if it is less than the current value, then Cont could produce a solution that improves upon the current value, and thus must be explored. The alternative Branch is disjoined to Cont, and DataCopy is restored to Data (ensuring that a future reset/3 copies the right variables);
  - if it is larger than or equal to the current value, then Cont can be safely discarded.
- 3. Finally, if the goal fails entirely, Min is the current minimum Value.

Nearest Neighbour Search The code in Figure 3 shows how the branch and bound framework efficiently solves the problem of finding the point (in a given set) that is nearest to a given target point on the Euclidean plane.

The run\_nn/3 predicate takes a point (X,Y), a Binary Space Partitioning (BSP)-tree<sup>5</sup> that represents the set of points, and returns the point, nearest to

<sup>&</sup>lt;sup>5</sup> A BSP-tree is a tree that recursively partitions a set of points on the Euclidean plane, by picking points and alternately splitting the plane along the x- or y-coordinate of those point. Splitting along the x-coordinate produces an xsplit/3 node, along the y-coordinate produces a ysplit/3 node.




Fig. 4. Nearest-Neighbour Search using a BSP-tree

(X,Y). The algorithm implemented by nn/3 recursively descends the BSP-tree. At each node it first tries the partition to which the target point belongs, then the point in the node, and finally the other partition. For this final step we can give an easy lower bound: any point in the other partition must be at least as far away as the (perpendicular) distance from the given point to the partition boundary.

As an example, we search for the point nearest to (1,0.1) in the set  $\{(0.5,0.5),(0,0),(-0.5,0),(-0.75,-0.5)\}$ . Figure 4 shows a BSP-tree containing these points, the solid lines demarcate the partitions. The algorithm visits the points (0.5,0.5) and (0,0), in that order. The shaded area is never visited, since the distance from (1,0.1) to the vertical boundary through (0,0) is greater than the distance to (0.5,0.5) (1 and about 0.64). The corresponding call to run\_nn/3 is:

#### 4.2 Probabilistic Programming

Probabilistic programming languages (PPLs) are programming languages designed for probabilistic modelling. In a probabilistic model, components behave in a variety of ways—just like in a non-deterministic model—but do so with a certain probability.

Instead of a single deterministic value, the execution of a probabilistic program results in a probability distribution of a set of values. This result is produced by probabilistic *inference* [15,6], for which there are many strategies and algorithms, the discussion of which is out of scope here. Here, we focus on one concrete probabilistic *logic* programming languages: PRISM [9].

A PRISM program consists of Horn clauses, and in fact, looks just like a regular Prolog program. However, we distinguish two special predicates:

```
msw(Key, Value) :- shift(msw(Key, Value)).
    prob(Goal) :-
2
         prob(Goal,ProbOut),
3
         write(Goal), write(': '), write(ProbOut), write('\n').
    prob(Goal,ProbOut) :-
         copy_term(Goal,GoalCopy),
         reset(GoalCopy, GoalCopy, Result),
         analyze_prob(GoalCopy,Result,ProbOut).
     analyze_prob(_,failure,0.0).
     analyze_prob(_,success(_,_),1.0).
10
     analyze_prob(_,shift(msw(K,V),C,_,Branch),ProbOut) :-
11
         values_x(K, Values, Probabilities),
         msw_prob(V,C,Values,Probabilities,0.0,ProbOfMsw),
13
         prob(Branch, BranchProb),
14
         ProbOut is ProbOfMsw + BranchProb.
```

Fig. 5. An implementation of probabilistic programming with delimited control.

- values\_x(Switch, Values, Probabilities) This predicate defines a probabilistic switch Switch, that can assume a value from Values with the probability that is given at the corresponding position in Probabilities (the contents of Probabilities should sum to one).
- msw(Switch, Value) This predicate samples a value Value from a switch Switch. For instance, if the program contains a switch declared as values\_x(coin, [h,t], [0.4,0.6]), then msw(coin,V) assigns h (for heads) to V with probability 0.4, and t (for tails) with probability 0.6. Remark that each distinct call to msw leads to a different sample from that switch. For instance, in the query msw(coin,X),msw(coin,Y), the outcome could be either (h,h),(t,t), (h,t) or (t,h).

Consider the following PRISM program, the running example for this section:

```
values_x(coin1,[h,t],[0.5,0.5]).
values_x(coin2,[h,t],[0.4,0.6]).
twoheads :- msw(coin1,h),msw(coin2,h).
onehead :- msw(coin1,V), (V = t, msw(coin2,h); V = h).
```

This example defines two predicates: twoheads which is true if both coins are heads, and onehead which is true if either coin is heads. However, note the special structure of onehead: PRISM requires the *exclusiveness condition*, that is, branches of a disjunction cannot be both satisfied at the same time. The simpler goal msw(coin1,heads); msw(coin2, heads) violates this assumption.

The code in Figure 5 interprets this program. Line 1 defines msw/2 as a simple shift. Lines 6–9 install a reset/3 call over the goal, and analyse the result. The result is analysed in the remaining lines: A failure never succeeds, and thus

has success probability 0.0 (line 9). Conversely, a successful computation has a success probability of 1.0 (line 10). Finally, the probability of a switch (lines 11-15) is the sum of the probability of the remainder of the program given each possible value of the switch multiplied with the probability of that value, and summed with the probability of the alternative branch.

The predicate msw\_prob finds the joint probability of all choices. It iterates over the list of values, and sums the probability of their continuations.

```
msw_prob(_,_,[],[],Acc,Acc).
msw_prob(V,C,[Value|Values],[Prob|Probs],Acc,ProbOfMsw) :-
prob((V = Value,C),ProbOut),
msw_prob(V,C,Values,Probs,Prob*ProbOut + Acc,ProbOfMsw).
```

Now, we can compute the probabilities of the two predicates above:

```
?- toplevel(prob(twoheads)).
twoheads: 0.25
?- toplevel(prob(onehead)).
onehead: 0.75
```

In Appendix B.3 we show how to implement the semantics a definite, non-looping fragment of ProbLog [6], another logic PPL, on top of the code in this section.

# 5 Properties of the Meta-Interpreter

In this section we establish two important correctness properties of our meta-interpreter. The proofs of these properties are in the corresponding Appendices A.1 and A.2 The first theorem establishes the soundness of the meta-interpreter, i.e., if a program (not containing shift/1 or reset/3) evaluates to success, then an SLD-derivation of the same answer must exist.

**Theorem 1 (Soundness)** For all lists of goals  $[A_1, \ldots, A_n]$ , terms  $\alpha, \beta, \gamma, \nu$ , variables P, R conjunctions  $B_1, \ldots, B_m$ ;  $C_1, \ldots, C_k$  and substitutions  $\theta$ , if

? 
$$-eval([A_1,\ldots,A_n],\alpha,alt(\beta,(B_1,\ldots,B_m)),P,R).$$
  
 $P = \nu, R = success(\gamma,C_1,\ldots,C_k).$ 

and the program contains neither shift/1 nor reset/3, then SLD-resolution<sup>6</sup> finds the following derivation:

$$\leftarrow (A_1, \dots, A_n, true); (\alpha = \beta, B_1, \dots, B_m)$$

$$\vdots$$

$$\Box$$
(with solution  $\theta$  s.t.  $\alpha \theta = \nu$ )

<sup>&</sup>lt;sup>6</sup> Standard SLD-resolution, augmented with disjunctions and conj/1 goals.

Conversely, we want to argue that the meta-interpreter is complete, i.e., if SLD-derivation finds a refutation, then meta-interpretation—provided that it terminates—must find the same answer eventually. The theorem is complicated somewhat by the fact that the first answer that the meta interpreter arrives at might not be the desired one due to the order of the clauses in the program. To deal with this problem, we use the operator ?-p, which is like ?-, but allows a different permutation of the program in every step.

**Theorem 2 (Completeness)** For all goals  $\leftarrow A_1, \ldots, A_n$ , if has solution  $\theta$ , then

?-p 
$$eval([A_1, \ldots, A_n], \alpha, alt(\beta, (B_1, \ldots, B_m)), P, R).$$
  
 $P = success(\gamma, (C_1, \ldots, C_k)), R = \alpha \theta.$ 

Together, these two theorems show that our meta-interpreter is a conservative extension of the conventional Prolog semantics.

#### 6 Related Work

Conjunctive Delimited Control Disjunctive delimited control is the culmination of a line of research on mechanisms to modify Prolog's control flow and search, which started with the hook-based approach of Tor [11] and was followed by the development of conjunctive delimited control for Prolog [10,12].

The listing below shows that disjunctive delimited control entirely subsumes conjunctive delimited control. The latter behaviour is recovered by disjoining the captured disjunctive branch. We believe that ToR is similarly superseded.

```
nd_reset(Goal,Ball,Cont) :-
   copy_term(Goal,GoalCopy),
   reset(GoalCopy,GoalCopy,R),
   ( R = failure -> fail
   ; R = success(BranchPattern,Branch) ->
      ( Goal = GoalCopy, Cont = 0
      ; Goal = BranchPattern, nd_reset(Branch,Ball,Cont))
   ; R = shift(X,C,BranchPattern,Branch) ->
      ( Goal = GoalCopy, Ball = X, Cont = C
      ; Goal = BranchPattern, nd_reset(Branch,Ball,Cont))
   ).
```

Abdallah [1] presents a higher-level interface for (conjunctive) delimited control on top of that of Schrijvers et al. [10]. In particular, it features *prompts*, first conceived in a Haskell implementation by Dyvbig et al. [4], which allow shifts to dynamically specify up to what reset to capture the continuation. We believe that it is not difficult to add a similar prompt mechanism on top of our disjunctive version of delimited control.

```
get(Interactor, Answer) :-
 get_engine(Interactor, Engine),
                                         % get engine state
 run_engine(Engine,NewEngine,Answer), % run up to the next answer
 update_engine(Interactor, NewEngine). % store the new engine state
return(X) :- shift(return(X)).
run_engine(engine(Pattern,Goal),NewEngine,Answer) :-
  reset(Pattern, Goal, Result),
 run engine result(Pattern, NewEngine, Answer, Result).
run_engine_result(Pattern,NewEngine,Answer,failure) :-
 NewEngine = engine(Pattern,fail),
  Answer
            = no.
run_engine_result(Pattern, NewEngine, Answer, success(BPattern, B)) :-
 NewEngine = engine(BPattern,B),
            = the(Pattern).
run_engine_result(Pattern,NewEngine,Answer,S) :-
 S = shift(return(X),C,BPattern,B)
 BPattern = Pattern,
 NewEngine = engine(Pattern,(C;B)),
  Answer
            = the(X).
```

Fig. 6. Interoperable Engines in terms of delimited control.

Interoperable Engines Tarau and Majumdar's Interoperable Engines [14] propose engines as a means for co-operative coroutines in Prolog. An engine is an independent instance of a Prolog interpreter that provides answers to the main interpreter on request.

The predicate new\_engine(Pattern,Goal,Interactor) creates a new engine with answer pattern Pattern that will execute Goal and is identified by Interactor. The predicate get(Interactor,Answer) has an engine execute its goal until it produces an answer (either by proving the Goal, or explicitly with return/1). After this predicate returns, more answers can be requested, by calling get/2 again with the same engine identifier. The full interface also allows bi-directional communication between engines, but that is out of scope here.

Figure 6 shows that we can implement the get/2 engine interface in terms of delimited control (the full code is available in Appendix B.2). The opposite, implementing disjunctive delimited control with engines, seems impossible as engines do not provide explicit control over the disjunctive continuation. Indeed, get/2 can only follow Prolog's natural left-to-right control flow and thus we cannot, e.g., run the disjunctive continuation before the conjunctive continuation, which is trivial with disjunctive delimited control.

Tabling without non-bactrackable variables Tabling [13,8] is a well-known technique that eliminates the sensitivity of SLD-resolution to clause and goal ordering, allowing a larger class of programs to terminate. As a bonus, it may improve the run-time performance (at the expense of increased memory consumption).

One way to implement tabling—with minimal engineering impact to the Prolog engine—is the tabling-as-a-library approach proposed by Desouter et al. [3]. This approach requires (global) mutable variables that are not erased by backtracking to store their data structures in a persistent manner. With the new reset/3 predicate, this is no longer needed, as (non-backtracking) state can be implemented in directly with disjunctive delimited control.

#### 7 Conclusion and Future Work

We have presented *disjunctive delimited control*, an extension to delimited control that takes Prolog's non-deterministic nature into account. This is a conservative extension that enables implementing disjunction-related language features and extensions as a library.

In future work, we plan to explore a WAM-level implementation of disjunctive delimited control, inspired by the stack freezing functionality of tabling engines, to gain access to the disjunctive continuations efficiently. Similarly, the use of copy\_term/2 necessitated by the current API has a detrimental impact on performance, which might be overcome by a sharing or shallow copying scheme.

#### References

- Abdallah, S.: More declarative tabling in Prolog using multi-prompt delimited control. CoRR abs/1708.07081 (2017)
- 2. Danvy, O., Filinski, A.: Abstracting control. pp. 151-160. LFP '90 (1990)
- 3. Desouter, B., van Dooren, M., Schrijvers, T.: Tabling as a library with delimited control. TPLP 15(4-5), 419–433 (2015)
- 4. Dyvbig, R.K., Jones, S.P., Sabry, A.: A monadic framework for delimited continuations. Tech. Rep. 615, Computer Science Department Indiana University (2005)
- Felleisen, M.: The theory and practice of first-class prompts. pp. 180–190. POPL '88 (1988)
- Fierens, D., den Broeck, G.V., Renkens, J., Shterionov, D.S., Gutmann, B., Thon, I., Janssens, G., Raedt, L.D.: Inference and learning in probabilistic logic programs using weighted boolean formulas. TPLP 15(3), 358–401 (2015)
- Saleh, A.H., Schrijvers, T.: Efficient algebraic effect handlers for Prolog. TPLP 16(5-6), 884–898 (2016)
- 8. Santos Costa, V., Rocha, R., Damas, L.: The YAP Prolog system. TPLP 12(1-2), 5–34 (2012)
- 9. Sato, T.: Generative modeling by PRISM. In: ICLP. LNCS, vol. 5649, pp. 24–35. Springer (2009)
- 10. Schrijvers, T., Demoen, B., Desouter, B., Wielemaker, J.: Delimited continuations for Prolog. TPLP **13**(4-5), 533–546 (2013)
- 11. Schrijvers, T., Demoen, B., Triska, M., Desouter, B.: Tor: Modular search with hookable disjunction. Sci. Comput. Program. **84**, 101–120 (2014)
- 12. Schrijvers, T., Wu, N., Desouter, B., Demoen, B.: Heuristics entwined with handlers combined: From functional specification to logic programming implementation. In: Proceedings of PPDP 2014. pp. 259–270. ACM (2014)

- 13. Swift, T., Warren, D.S.: XSB: Extending Prolog with tabled logic programming. TPLP **12**(1-2), 157–187 (Jan 2012)
- 14. Tarau, P., Majumdar, A.K.: Interoperating logic engines. In: Gill, A., Swift, T. (eds.) PADL 2009 Proceedings. LNCS, vol. 5418, pp. 137–151. Springer (2009)
- 15. Wood, F.D., van de Meent, J., Mansinghka, V.: A new approach to probabilistic programming inference. In: AISTATS. JMLR Workshop and Conference Proceedings, vol. 33, pp. 1024–1032. JMLR.org (2014)

# Appendix Table of Contents

| A            | Corr | ectness Proofs              | 8  |
|--------------|------|-----------------------------|----|
|              | A.1  | Evaluation Is Sound         | 8  |
|              | A.2  | Evaluation Is Complete      | 24 |
| В            | Add  | itional Examples            | )[ |
|              | B.1  | Negation                    | 25 |
|              | B.2  | Interoperable Engines       | 25 |
|              | B.3  | ProbLog 2                   | 26 |
| $\mathbf{C}$ | Exai | mple Meta-Interpreter Trace | 35 |
| D            | Full | Meta-Interpreter            | 30 |

#### A Correctness Proofs

#### A.1 Evaluation Is Sound

**Theorem 1 (Soundness)** For all lists of goals  $[A_1, \ldots, A_n]$ , terms  $\alpha, \beta, \gamma, \nu$ , variables P, R conjunctions  $B_1, \ldots, B_m$ ;  $C_1, \ldots, C_k$  and substitutions  $\theta$ , if

?- 
$$eval([A_1, \ldots, A_n], \alpha, alt(\beta, (B_1, \ldots, B_m)), P, R).$$
  
 $P = \nu, R = success(\gamma, C_1, \ldots, C_k).$ 

and the program contains neither shift/1 nor reset/3, then SLD-resolution<sup>7</sup> to finds the following derivation:

$$\leftarrow (A_1, \dots, A_n, true); (\alpha = \beta, B_1, \dots, B_m)$$

$$\vdots$$

$$(with solution \theta s.t. \alpha\theta = \nu)$$

*Proof.* By induction on the Prolog derivation.

Case [] In this case the only possible result is  $P = \alpha$ :

?- 
$$eval([], \alpha, alt(\beta, (B_1, \dots, B_m)), P, R)$$
.  $\Longrightarrow \begin{cases} \leftarrow true; (\alpha = \beta, B_1, \dots, B_m) \\ \leftarrow true (= \square) \end{cases}$  (with solution  $\epsilon$ )

where  $\epsilon$  is the empty substitution.

 $\it Case\ [true/{\tt Conj}]$  In this case the body contains only a direct recursive call, we have:

$$?- \operatorname{eval}([\operatorname{true}, A_2, \dots, A_n], \alpha, \operatorname{alt}(\beta, (B_1, \dots, B_m)), P, R).$$

$$P = \nu, R = \operatorname{success}(\gamma, (C_1, \dots, C_k)).$$

$$?- \operatorname{eval}([A_2 \dots, A_n], \alpha, \operatorname{alt}(\beta, (B_1, \dots, B_m)), P, R).$$

$$P = \nu, R = \operatorname{success}(\gamma, (C_1, \dots, C_k)).$$

$$(A_2, \dots, A_n, \operatorname{true}); (\alpha = \beta, B_1, \dots, B_m)$$

$$\vdots$$

$$(\operatorname{with solution} \theta \text{ s.t. } \alpha\theta = \nu)$$

$$\leftarrow (\operatorname{true}, A_2, \dots, A_n, \operatorname{true}); (\alpha = \beta, B_1, \dots, B_m)$$

$$\vdots$$

$$(\operatorname{with solution} \theta \text{ s.t. } \alpha\theta = \nu)$$

$$(\operatorname{with solution} \theta \text{ s.t. } \alpha\theta = \nu)$$

$$(\operatorname{with solution} \theta \text{ s.t. } \alpha\theta = \nu)$$

 $<sup>^7</sup>$  Standard SLD-resolution, augmented with disjunctions and conj/1 goals.

Case [(G1,G2)/CONJ] A straightforward calculation gives the desired result:

$$?- \operatorname{eval}([(G_1, G_2), A_2, \dots, A_n], \alpha, \operatorname{alt}(\beta, (B_1, \dots, B_m)), P, R).$$

$$P = \nu, R = \operatorname{success}(\gamma, (C_1, \dots, C_k)).$$

$$\boxed{ \textbf{Definition and Determinism of eval/5}}$$

$$?- \operatorname{eval}([G_1, G_2, A_2, \dots, A_n], \alpha, \operatorname{alt}(\beta, (B_1, \dots, B_m)), P, R).$$

$$P = \nu, R = \operatorname{success}(\gamma, (C_1, \dots, C_k)).$$

$$\leftarrow (G_1, G_2, A_2, \dots, A_n, \operatorname{true}); (\alpha = \beta, B_1, \dots, B_m)$$

$$\vdots \qquad \qquad \boxed{ \textbf{Induction} }$$

$$(\text{with solution } \theta \text{ s.t. } \alpha\theta = \nu)$$

$$\leftarrow ((G_1, G_2), A_2, \dots, A_n, \operatorname{true}); (\alpha = \beta, B_1, \dots, B_m)$$

$$\vdots \qquad \qquad \boxed{ \textbf{Associativity} }$$

$$(\text{with solution } \theta \text{ s.t. } \alpha\theta = \nu)$$

Case [fail/Conj] This case immediately calls backtrack/3:

$$?- eval([fail, A_2, \dots, A_n], \alpha, alt(\beta, (B_1, \dots, B_m)), P, R).$$

$$P = \nu, R = success(\gamma, (C_1, \dots, C_k)).$$

$$?- backtrack(alt(\beta, (B_1, \dots, B_m)), P, R).$$

$$P = \nu, R = success(\gamma, (C_1, \dots, C_k)).$$

$$P = \nu, R = success(\gamma, (C_1, \dots, C_k)).$$

$$Definition and Determinism of eval/5$$

At this point, we can see that  $alt(\beta, (B_1, ..., B_m))$  cannot be empty, for otherwise R = failure. Execution of the else branch then gives:

Case [(G1;G2),CONJ] Assume that the result of the copy\_term/2 call is  $alt(\delta, conj([D_1, \ldots, D_n]))$ , where all the variables are fresh, such that

$$\delta \sigma = \alpha \sigma \Rightarrow (D_1, \dots, D_n) \sigma = (G_2, A_2, \dots, A_n) \sigma$$
 (1)

Then the result of disjoining this with  $alt(\beta, (B_1, \ldots, B_m))$  is the new disjunction

$$\xi = alt(F, ((\delta = F, conj([D_1, \dots, D_n])); (\beta = F, B_1, \dots, B_m)))$$

with F fresh. Now we reason as follows:

?- 
$$eval([(G_1; G_2)A_2, \dots, A_n], \alpha, alt(\beta, (B_1, \dots, B_m)), P, R).$$
 $P = \nu, R = success(\gamma, (C_1, \dots, C_k)).$ 
?-  $eval([G_1, A_2, \dots, A_n], \alpha, \xi, P, R).$ 
 $P = \nu, R = success(\gamma, (C_1, \dots, C_k)).$ 

Definition and Determinism of eval/5
$$\leftarrow (G_1, A_2, \dots, A_n, true); (\alpha = F, ((\delta = F, conj([D_1, \dots, D_n])); (\beta = F, B_1, \dots, B_m))))$$

$$\Rightarrow \qquad \vdots \qquad \qquad \vdots \qquad \qquad (\text{with solution } \theta \text{ s.t. } \alpha\theta = \nu)$$

Associativity, Distributivity, resolution of conj/1,  $F$  is local variable
$$\leftarrow (G_1, A_2, \dots, A_n, true); (\alpha = \delta, D_1, \dots, D_n); (\alpha, \beta, B_1, \dots, B_m)$$

$$\Rightarrow \qquad \vdots \qquad \qquad \vdots \qquad \qquad \vdots$$

$$(\text{with solution } \theta \text{ s.t. } \alpha\theta = \nu)$$

Note that if  $\sigma$  refutes  $\leftarrow \alpha = \delta, D_1, \dots, D_n$ ; then  $\delta \sigma = \alpha \sigma$ , hence (1)  $(D_1, \dots, D_n) \sigma = (G_2, A_2, \dots, A_n) \sigma$ . Thus,  $\sigma$  also refutes  $\leftarrow G_2, A_2, \dots, A_n$ .

So, we may write the above as:

$$\leftarrow (G_1, A_2, \dots, A_n, true); (G_2, A_2, \dots, A_n); (\alpha = \beta, B_1, \dots, B_m)$$

$$\vdots$$

$$(\text{with solution } \theta \text{ s.t. } \alpha \theta = \nu)$$

$$\boxed{\textbf{Distributivity}}$$

$$\leftarrow ((G_1; G_2), A_2, \dots, A_n, true); (\alpha = \beta, B_1, \dots, B_m)$$

$$\vdots$$

$$(\text{with solution } \theta \text{ s.t. } \alpha \theta = \nu)$$

Case [conj(Cs)/Conj]

?- 
$$eval([conj([D_1, ..., D_l]), A_2, ..., A_n], \alpha, alt(\beta, (B_1, ..., B_m)), P, R).$$
 $P = \nu, R = success(\gamma, (C_1, ..., C_k)).$ 

Definition and Determinism of eval/5

?-  $eval([D_1, ..., D_l, A_2, ..., A_n], \alpha, alt(\beta, (B_1, ..., B_m)), P, R).$ 
 $P = \nu, R = success(\gamma, (C_1, ..., C_k)).$ 

Induction

 $\leftarrow (D_1, ..., D_l, A_2, ..., A_n, true); (\alpha = \beta, B_1, ..., B_m)$ 
 $\vdots$ 

(with solution  $\theta$  s.t.  $\alpha \theta = \nu$ )

Induction

 $\leftarrow (conj([D_1, ..., D_l]), A_2, ..., A_n, true); (\alpha = \beta, B_1, ..., B_m)$ 
 $\vdots$ 
 $\vdots$ 

(with solution  $\theta$  s.t.  $\alpha \theta = \nu$ )

Case [C/CONJ] In this case there are two possibilities: if no clauses match with the head, Clauses is empty, and the procedure backtracks. In this case the reasoning is identical to the case [fail|CONJ].

Otherwise, assume we have a list  $[(H :-Body)_1, \ldots, (H :-Body)_l]$  of clauses that match C. After disjoin\_clauses/3, this list becomes a disjunction  $(C = H_1, Body_1); \cdots; (C = H_l, Body_l)$ .

From the recursive call we get:

$$?- \operatorname{eval}([C|A_2, \cdots, A_n], \alpha, \operatorname{alt}(\beta, (B_1, \dots, B_m)), P, R).$$

$$P = \nu, R = \operatorname{result}(\gamma, (C_1, \dots, C_k))$$

$$\boxed{ \textbf{Definition and Determinism of eval/5}}$$

$$?- \operatorname{eval}([(C = H_1, \operatorname{Body_1}); \dots; (C = H_l, \operatorname{Body_l})|A_2, \dots, A_n], \alpha, \operatorname{alt}(\beta(B_1, \dots, B_m)), P, R).$$

$$P = \nu, R = \operatorname{result}(\gamma(C_1, \dots, C_k))$$

$$\boxed{ \textbf{Induction}}$$

$$\leftarrow ((C = H_1, \operatorname{Body_1}); \dots; (C = H_l, \operatorname{Body_l})), A_2, \dots, A_n, \operatorname{true}); (\alpha = \beta, B_1, \dots, B_m)$$

$$\vdots$$

$$(\text{with solution } \theta, \text{ s.t. } \alpha\theta = \nu)$$

If  $\leftarrow ((C = H_1, Body_1); \cdots; (C = H_l, Body_l)), A_2, \dots, A_n, true)$  has a refutation  $\theta$ , then let  $\theta = \theta_i \theta'$  such that  $\theta_i$  refutes  $\leftarrow C = H_i, Body_i$  and  $\theta'$  refutes

$$\leftarrow A_2, \ldots, A_n$$
:

Then  $\theta$  also refutes  $\leftarrow C, A_2, \dots, A_n$ , so we can conclude:

$$\leftarrow (C, A_2, \dots, A_n, true); (\alpha = \beta, B_1, \dots, B_m)$$

$$\vdots$$

$$\Box$$
(with solution  $\theta$ , s.t.  $\alpha \theta = \nu$ )

#### A.2 Evaluation Is Complete

**Theorem 2** For all goals  $\leftarrow A_1, \ldots, A_n$ , if

$$\leftarrow A_1, \dots, A_n$$
 $\vdots$ 
 $\Box$ 
(with solution  $\theta$ )

then

?-p 
$$eval([A_1, \ldots, A_n], \alpha, alt(\beta, (B_1, \ldots, B_m)), P, R).$$
  
 $P = success(\gamma, (C_1, \ldots, C_k)), R = \alpha \theta.$ 

where ?- $_p$  is defined as ?-, but each step is allowed to use a different permutation of the program.

*Proof.* By induction on the length of the derivation.

Base 
$$n = 0$$
:  $\leftarrow A_1, \dots, A_n = \leftarrow true = \square$  then  $\theta$  is empty, and 
$$?\text{-}_{p} \ eval([], \alpha, alt(\beta, (B_1, \dots, B_m)), P, R).$$

$$P = success(\beta, (B_1, \dots, B_m)), R = \alpha$$

by definition.

Induction Let

$$\leftarrow A_1, A_2, \dots, A_n$$

$$\leftarrow G_1, \dots, G_f, A_2, \dots, A_n \theta_1 \text{ (clause } H \leftarrow G_1, \dots, G_f; A_1 \theta_1 = H \theta_1)$$

$$\vdots$$

$$\Box$$
(with substitution  $\theta = \theta_1 \theta_2$ )

By induction we have,

$$\begin{array}{l} ?\text{-}_{\mathbf{p}} \ eval([G_1,\ldots,G_f,A_2,\ldots,A_n\theta_1],\alpha\theta_1, alt(\delta,(D_1,\ldots,D_l)),P,R) \\ P = success(\gamma,(C_1,\ldots,C_k)),S = \alpha\theta_1\theta_2. \\ \\ \Leftrightarrow ?\text{-}_{\mathbf{p}} \ eval([(A_1 = H,G_1,\ldots,G_f;\ldots),A_2,\ldots,A_n],\alpha, alt(\beta,(B_1,\ldots,B_m)),P,R) \\ P = success(\gamma,(C_1,\ldots,C_k)),S = \alpha\theta_1\theta_2. \\ \\ \Rightarrow ?\text{-}_{\mathbf{p}} \ eval([A_1,A_2,\ldots,A_n],\alpha, alt(\beta,(B_1,\ldots,B_m)),P,R) \\ P = success(\gamma,(C_1,\ldots,C_k)),S = \alpha\theta_1\theta_2. \end{array}$$

## B Additional Examples

#### B.1 Negation

G = GC,

```
not(G) :-
  copy_term(G,GC),
  reset(GC,GC,R),
  R = failure.
B.2 Interoperable Engines
engines(G) :-
    engines(G,[]).
new_engine(Pattern,Goal,Interactor) :-
    shift(new_engine(Pattern,Goal,Interactor)).
get(Interactor, Answer) :-
    shift(get(Interactor, Answer)).
return(X) :-
    shift(return(X)).
engines(G,EngineList) :-
    copy term(G,GC),
    reset(GC,GC,R),
    engines_result(G,GC,EngineList,R).
engines_result(_,_,_,failure) :-
    fail.
engines result(G,GC,EngineList,success(BC,B)) :-
    (G = GC; G = BC, engines(B, EngineList)).
engines_result(G,GC,EngineList,S) :-
    S = shift(new_engine(Pattern, Goal, Interactor), C, BC, B),
    length(EngineList,Interactor),
    copy_term(Pattern-Goal, PatternCopy-GoalCopy),
    NewEngineList = [Interactor-engine(PatternCopy, GoalCopy) | EngineList],
    G = GC,
    G = BC
    engines((C;B),NewEngineList).
engines_result(G,GC,EngineList,S) :-
    S = shift(get(Interactor, Answer), C, BC, B),
    member(Interactor-Engine, EngineList),
    run_engine(Engine, NewEngine, Answer),
    update(Interactor, NewEngine, EngineList, NewEngineList),
```

```
G = BC
    engines((C;B),NewEngineList).
update(K, NewV, [K-_|T], [K-NewV|T]).
update(K,NewV,[OtherK-V|T],[OtherK-V|T2]) :-
    K \== OtherK,
    update(K, NewV, T, T2).
run_engine(engine(Pattern,Goal),NewEngine,Answer) :-
    reset(Pattern,Goal,Result),
    run_engine_result(Pattern, NewEngine, Answer, Result).
run engine result(Pattern, NewEngine, Answer, failure) :-
    NewEngine = engine(Pattern, fail),
              = no.
run_engine_result(Pattern,NewEngine,Answer,success(BPattern,B)) :-
    NewEngine = engine(BPattern,B),
    Answer
            = the(Pattern).
run_engine_result(Pattern,NewEngine,Answer,shift(return(X),C,BPattern,B)) :-
    BPattern = Pattern,
    NewEngine = engine(Pattern,(C;B)),
            = the(X).
    Answer
B.3 ProbLog
fact(F) :-
    shift(fact(F,V)),
    V = t.
is_true(F,Pc) :- member(F-t,Pc).
is_false(F,Pc) :-member(F-f,Pc).
problog(Goal) :- problog(Goal,[]).
problog(Goal,Pc) :-
    reset(Goal,Goal,Result),
    analyze_problog(Result,Pc).
analyze_problog(success(_,_),_Pc).
analyze_problog(shift(fact(F,V),C,_,Branch),Pc) :-
    is_true(F,Pc),
    V = t,
    problog((C;Branch),Pc).
analyze_problog(shift(fact(F,V),C,_,Branch),Pc) :-
    is_false(F,Pc),
    V = f,
    problog((C;Branch),Pc).
```

```
analyze_problog(shift(fact(F,V),C,_,Branch),Pc) :-
   not(is_true(F,Pc)),
   not(is_false(F,Pc)),
   msw(F,V),
   problog((C;Branch),[F-V|Pc]).
analyze_problog(failure,_Pc) :- fail.
% NEGATION DOES NOT WORK.
  Example usage:
 % 0.5 :: f1.
 values_x(f1,[t,f],[0.5,0.5]).
 f1 :- fact(f1).
 % 0.5 :: f2.
 values_x(f2,[t,f],[0.5,0.5]).
 f2 :- fact(f2).
 p :- f1.
 p :- f2
 ?- solutions(prob(problog((f1,f1)))).
 problog((f1,f1)): 0.5
 ?- solutions(prob(problog(p))).
 problog(p): 0.75
```

# C Example Meta-Interpreter Trace

Consider the simple program:

```
p(1).
p(2) :- shift(2).
```

The following table shows the values of each of the arguments of the meta-interpreter, while evaluating the goal p(X).

| PatI  | n Conj          | Disj                                     |  |  |  |  |  |
|---|-----------------|--|--|--|--|--|--|
| Х   | [p(X)]          | alt(Y,fail)                              |  |  |  |  |  |
| Х   | [(X=1,true;     | alt(Y,fail)                              |  |  |  |  |  |
|   | X = 2, shift(2) | ]  |  |  |  |  |  |
| X   | [(X=1,true)]    | alt(Z1,(Z1=X1,X1=2,shift(2); Z1=Y,fail)) |  |  |  |  |  |
| X   | [X=1,true]      | alt(Z1,(Z1=X1,X1=2,shift(2); Z1=Y,fail)) |  |  |  |  |  |
| 1   | [true]          | alt(Z1,(Z1=X1,X1=2,shift(2); Z1=Y,fail)) |  |  |  |  |  |
| 1   | []              | alt(Z1,(Z1=X1,X1=2,shift(2); Z1=Y,fail)) |  |  |  |  |  |
| PatOut=1, Result=success(Z1,(Z1=X1,X1=2,shift(2); Z1=Y,fail)) |                 |  |  |  |  |  |  |

We can then evaluate the alternative branches:

| PatIn Conj Disj   |                         |              |                                   |  |  |  |  |
|---|-------------------------|--------------|-----------------------------------|--|--|--|--|
| Z1  | [(Z1=X1,X1=2,shift(2);  | Z1=Y,fail))] | alt(A,fail)                       |  |  |  |  |
| Z1  | [(Z1=X1,X1=2,shift(2))] |              | alt(B,(Z2=B,Z2=Y,fail;            |  |  |  |  |
|   |                         |              | A=B,fail)                         |  |  |  |  |
| Z1  | [Z1=X1,X1=2,shift(2)]   |              | <pre>alt(B,(Z2=B,Z2=Y,fail;</pre> |  |  |  |  |
|   |                         |              | A=B,fail)                         |  |  |  |  |
| X1  | [X1=2,shift(2)]         |              | <pre>alt(B,(Z2=B,Z2=Y,fail;</pre> |  |  |  |  |
|   |                         |              | A=B,fail)                         |  |  |  |  |
| 2   | [shift(2))              |              | <pre>alt(B,(Z2=B,Z2=Y,fail;</pre> |  |  |  |  |
|   |                         |              | A=B,fail)                         |  |  |  |  |
| 2   |                         |              | <pre>alt(B,(Z2=B,Z2=Y,fail;</pre> |  |  |  |  |
|   |                         |              | A=B,fail)                         |  |  |  |  |
| <pre>Result=shift(2,conj([]),B,(Z2=B,Z2=Y,fail;A=B,fail))</pre> |                         |              |                                   |  |  |  |  |

and again

# D Full Meta-Interpreter

The following code has been tested on SWI-Prolog versions 7.6.3 and 8.0.3. It uses the type\_check package to type-check the Prolog code. The type-checking annotations can be commented out, without loss of functionality on systems that do not support type-checking.

```
%%-----
%% Meta-interpreter for disjunctive delimited control.
%% The meta-interpreter supports most basic Prolog features, except if then
%% else.
%%
%% Author: Alexander Vandenbroucke (alexander.vandenbroucke@kuleuven.be)
%% Author: Tom Schrijvers (tom.schrijvers@kuleuven.be)
:- style_check(-singleton).
:- use_module(library(type_check)).
%% preliminaries
% The operator :=/2 is a more strictly typed synonym of =/2.
:- op(700,xfx,:=).
:- pred := (A,A).
:=(X,Y):-X=Y.
:- trust_pred copy_term(A,A).
:- trust pred clause(pred, pred).
:- trust_pred shift(any).
:- trust_pred reset(P,pred,result(P)).
:- trust_pred write(any).
:- trust_pred length(list(A),integer).
:- trust_pred member(A,list(A)).
%%------
%% all solutions
:- pred solutions(pred).
solutions(G0) :-
   copy term(GO,G),
   copy_term(G,GC),
   empty_disj(EmptyDisj),
```

```
do_reset((GC,true),G,GC,Result,EmptyDisj),
    analyze_solutions(GO,G,Result).
:- pred analyze_solutions(pred,pred,result(pred)).
analyze_solutions(GO,G,success(PatternCopy,Branch)) :-
    (GO = G
    ; copy_term(PatternCopy,PC),
      empty_disj(EmptyDisj),
      do_reset((Branch,true),PC,PatternCopy,Result,EmptyDisj),
      analyze_solutions(G0,PC,Result)
    ).
analyze_solutions(_,_,shift(_,_,_,_)) :-
    write('solutions: unexpected shift/1.\n'),
analyze_solutions(_,_,failure) :- fail.
%% Meta Interpreter
:- pred do_reset(pred,P,P,result(P),disjunction(P)).
%% do_reset(Conjuction,Pattern,PatternCopy,Result,Disjunction)
\% * On success or shift, Pattern contains the current solution.
   Pattern MUST be an unbound variable.
%% * PatternCopy is a fresh copy of Pattern. Note: the contents of PatternCopy
   are not always sensible after do_reset/5, hence PatternCopy should not be
%%
    inspected afterwards.
%% * Afterwards Result is one of:
%%
      - success(BranchPattern, Branch): BranchPattern is a fresh copy of
        Pattern, Branch is a disjunctive continuation (see below). In this case
%%
        PatternCopy and Pattern are unified, to carry out the current
%%
        substitution.
      - shift(Term, Conjunction, BranchPattern, Branch):
        Term is the term that was shifted, Conjuction is the conjunctive
%%
        continuation. BranchPattern is a fresh copy of Pattern, Branch is the
%%
        disjunctive continuation. Pattern and PatternCopy are unified to carry
%%
        out the current substitution.
      - failure: when the conjunction and the disjunction fail
     Result MUST be an unbound variable.
1 contains %% * Conjunction: the current conjunctive goal. The variables it contains
     should not occur in Pattern. Moreover, the final conjunct should always be
%%
     true/O. The conjunction must NOT be empty.
%% * Disjunction: a disjunction/2, that contains the disjunctive goal (see
```

```
%%
     empty_disj/1, disjoin/3 below.
%% single true/0-pattern (base case)
do_reset(true,Pattern,PatternCopy,Result,Disj) :-
    disjunction(BranchPattern, Branch) := Disj,
    Pattern := PatternCopy,
    Result := success(BranchPattern, Branch).
%% fail/0-pattern
do_reset((fail,_Conj),Pattern,_,Result,Disj) :-
    !,
    backtrack(Pattern, Result, Disj).
%% true/0-pattern
do_reset((true,Conj),Pattern,PatternCopy,Result,Disj) :-
    do_reset(Conj,Pattern,PatternCopy,Result,Disj).
%% conjunction pattern
do_reset(((G1,G2),Conj),Pattern,PatternCopy,Result,Disj) :-
    do_reset((G1,(G2,Conj)),Pattern,PatternCopy,Result,Disj).
%% disjunction pattern
do_reset(((G1;G2),Conj),Pattern,PatternCopy,Result,Disj) :-
    copy_term(disjunction(PatternCopy,(G2,Conj)),Branch),
    disjoin(Branch, Disj, New Disj),
    do reset((G1,Conj),Pattern,PatternCopy,Result,NewDisj).
%% shift/1-pattern
do_reset((shift(X),Conj),Pattern,PatternCopy,Result,Disj) :-
    !,
    Pattern := PatternCopy,
           := disjunction(BranchPattern, Branch),
    Disj
    Result := shift(X,Conj,BranchPattern,Branch).
%% (new) reset/3-pattern
do_reset((reset(P,G,R),Conj),Pattern,PatternCopy,Result,Disj) :-
   copy_term(P-G,PC-GC),
    empty_disj(D),
    do_reset((GC,true),Q,PC,S,D),
    type_to_any(R,RAny),type_to_any(S,SAny),
```

```
do_reset((P = Q, (RAny = SAny, Conj)),Pattern,PatternCopy,Result,Disj).
%% unification pattern
do_reset((X = Y,Conj),Pattern,PatternCopy,Result,Disj) :-
    ( X = Y -> do_reset(Conj,Pattern,PatternCopy,Result,Disj)
    ; backtrack(Pattern, Result, Disj)).
%% unification pattern
do_reset((X = Y,Conj),Pattern,PatternCopy,Result,Disj) :-
    ( X = Y -> do_reset(Conj,Pattern,PatternCopy,Result,Disj)
    ; backtrack(Pattern, Result, Disj)).
%%term non-equivalence pattern
do_reset((X \== Y,Conj),Pattern,PatternCopy,Result,Disj) :-
    ( X \== Y -> do_reset(Conj,Pattern,PatternCopy,Result,Disj)
    ; backtrack(Pattern, Result, Disj)).
%% copy_term/2-pattern
do_reset((copy_term(X,Y),Conj),Pattern,PatternCopy,Result,Disj) :-
    !,
    ( copy_term(X,Y) -> do_reset(Conj,Pattern,PatternCopy,Result,Disj)
    ; backtrack(Pattern, Result, Disj)).
%% cut/0-pattern
do_reset((!,Conj),Pattern,PatternCopy,Result,Disj) :-
    do reset(Conj,Pattern,PatternCopy,Result,Disj).
%% findall/3-pattern
do_reset((findall(T,G,L),Conj),Pattern,PatternCopy,Result,Disj) :-
    ( findall(T,G,L) -> do_reset(Conj,Pattern,PatternCopy,Result,Disj)
    ; backtrack(Pattern, Result, Disj)).
%% length/2-pattern
do_reset((length(L,X),Conj),Pattern,PatternCopy,Result,Disj) :-
    ( length(L,X) -> do_reset(Conj,Pattern,PatternCopy,Result,Disj)
    ; backtrack(Pattern, Result, Disj)).
%% member/2-pattern
do_reset((member(X,L),Conj),Pattern,PatternCopy,Result,Disj) :-
```

```
( member(X,L) -> do_reset(Conj,Pattern,PatternCopy,Result,Disj)
    ; backtrack(Pattern, Result, Disj)).
%% is/2-pattern
do_reset((is(R,Exp),Conj),Pattern,PatternCopy,Result,Disj) :-
    !,
    ( R is Exp -> do_reset(Conj,Pattern,PatternCopy,Result,Disj)
    ; backtrack(Pattern, Result, Disj)).
%% (<)/2-pattern
do reset((<(X,Y),Conj),Pattern,PatternCopy,Result,Disj) :-</pre>
    ( X < Y -> do_reset(Conj,Pattern,PatternCopy,Result,Disj)
    ; backtrack(Pattern, Result, Disj)).
%% (>=)/2-pattern
do_reset((>=(X,Y),Conj),Pattern,PatternCopy,Result,Disj) :-
    ( X >= Y -> do_reset(Conj,Pattern,PatternCopy,Result,Disj)
    ; backtrack(Pattern, Result, Disj)).
%% (@<)/2-pattern
do_reset((@<(X,Y),Conj),Pattern,PatternCopy,Result,Disj) :-</pre>
    ( X @< Y -> do_reset(Conj,Pattern,PatternCopy,Result,Disj)
    ; backtrack(Pattern, Result, Disj)).
%% (@>=)/2-pattern
do reset((@>=(X,Y),Conj),Pattern,PatternCopy,Result,Disj) :-
    ( X @>= Y -> do_reset(Conj,Pattern,PatternCopy,Result,Disj)
    ; backtrack(Pattern, Result, Disj)).
%% write/1-pattern
do_reset((write(T),Conj),Pattern,PatternCopy,Result,Disj) :-
    !,
    do_reset(Conj,Pattern,PatternCopy,Result,Disj).
%% clause pattern.
do_reset((G,Conj),Pattern,PatternCopy,Result,Disj) :-
    findall(GC-Body,(clause(G,Body), GC = G),Clauses),
    ( Clauses := [] ->
```

```
backtrack(Pattern, Result, Disj)
    ; disjoin_clauses(G,Clauses,ClausesDisj),
      do_reset((ClausesDisj,Conj),Pattern,PatternCopy,Result,Disj)
    ).
:- pred disjoin_clauses(P,list(pair(P,pred)),pred).
disjoin_clauses(_,[],fail).
% The next clause is not necessary, but makes things prettier when tracing.
disjoin_clauses(G,[GC-Clause],(G=GC,Clause)) :- !.
disjoin_clauses(G,[GC-Clause|Clauses], ((G=GC,Clause); Disj) ) :-
    disjoin_clauses(G,Clauses,Disj).
:- pred backtrack(P,result(P),disjunction(P)).
%% backtracking
backtrack(Pattern,Result,Disj) :-
    ( empty_disj(Disj) ->
         Result := failure
    ; Disj := disjunction(PatternCopy,G) ->
         empty_disj(EmptyDisj),
         do_reset((G,true),Pattern,PatternCopy,Result,EmptyDisj)
         %% TODO: could be more efficient if we pattern match on G?
    ).
%% Disjunctions
%%
"My Disjunctions are of the from disjunction(Pattern, Goal), where Goal may
%% contain a disjunction.
%% Disjunctions are neither commutative nor associative. However empty_disj/1
%% is still the unit for disjoin/3.
:- type disjunction(Pattern) ---> disjunction(Pattern, pred).
%% The empty disjunction.
:- pred empty_disj(disjunction(Pattern)).
empty_disj(disjunction(_,fail)).
%% Disjoin two disjunctions.
%% This operation is not commutative
:- pred disjoin(disjunction(P), disjunction(P), disjunction(P)).
disjoin(disjunction(_,fail),Disjunction,Disjunction) :- !.
disjoin(Disjunction, disjunction(_,fail), Disjunction) :- !.
```